

NAG Library Routine Document

F04BAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F04BAF computes the solution to a real system of linear equations $AX = B$, where A is an n by n matrix and X and B are n by r matrices. An estimate of the condition number of A and an error bound for the computed solution are also returned.

2 Specification

```
SUBROUTINE F04BAF(N, NRHS, A, LDA, IPIV, B, LDB, RCOND, ERBND, IFAIL)
INTEGER          N, NRHS, LDA, IPIV(N), LDB, IFAIL
double precision A(LDA,*), B(LDB,*), RCOND, ERBND
```

3 Description

The LU decomposition with partial pivoting and row interchanges is used to factor A as $A = PLU$, where P is a permutation matrix, L is unit lower triangular, and U is upper triangular. The factored form of A is then used to solve the system of equations $AX = B$.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

5 Parameters

- 1: N – INTEGER *Input*
On entry: the number of linear equations n , i.e., the order of the matrix A .
Constraint: $N \geq 0$.
- 2: NRHS – INTEGER *Input*
On entry: the number of right-hand sides r , i.e., the number of columns of the matrix B .
Constraint: $NRHS \geq 0$.
- 3: A(LDA,*) – **double precision** array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the n by n coefficient matrix A .
On exit: if $IFAIL \geq 0$, the factors L and U from the factorization $A = PLU$. The unit diagonal elements of L are not stored.

- 4: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F04BAF is called.
Constraint: $LDA \geq \max(1, N)$.
- 5: IPIV(N) – INTEGER array *Output*
On exit: if $IFAIL \geq 0$, the pivot indices that define the permutation matrix P ; at the i th step row i of the matrix was interchanged with row $IPIV(i)$. $IPIV(i) = i$ indicates a row interchange was not required.
- 6: B(LDB,*) – **double precision** array *Input/Output*
Note: the second dimension of the array B must be at least $\max(1, NRHS)$. To solve the equations $Ax = b$, where b is a single right-hand side, B may be supplied as a one-dimensional array with length $LDB = \max(1, N)$.
On entry: the n by r matrix of right-hand sides B .
On exit: if $IFAIL = 0$ or $N + 1$, the n by r solution matrix X .
- 7: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which F04BAF is called.
Constraint: $LDB \geq \max(1, N)$.
- 8: RCOND – **double precision** *Output*
On exit: if $IFAIL \geq 0$, an estimate of the reciprocal of the condition number of the matrix A , computed as $RCOND = 1 / (\|A\|_1 \|A^{-1}\|_1)$.
- 9: ERRBND – **double precision** *Output*
On exit: if $IFAIL = 0$ or $N + 1$, an estimate of the forward error bound for a computed solution $\hat{x}$, such that $\|\hat{x} - x\|_1 / \|x\|_1 \leq ERRBND$, where $\hat{x}$ is a column of the computed solution returned in the array B and x is the corresponding column of the exact solution X . If RCOND is less than **machine precision**, then ERRBND is returned as unity.
- 10: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: $IFAIL = 0$ unless the routine detects an error (see Section 6).
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL < 0$ and $IFAIL \neq -999$

If $IFAIL = -i$, the i th argument had an illegal value.

IFAIL = -999

Allocation of memory failed. The INTEGER allocatable memory required is N , and the **double precision** allocatable memory required is $4 \times N$. In this case the factorization and the solution X have been computed, but RCOND and ERRBND have not been computed.

IFAIL > 0 and IFAIL ≤ N

If IFAIL = i , u_{ii} is exactly zero. The factorization has been completed, but the factor U is exactly singular, so the solution could not be computed.

IFAIL = $N + 1$

RCOND is less than **machine precision**, so that the matrix A is numerically singular. A solution to the equations $AX = B$ has nevertheless been computed.

7 Accuracy

The computed solution for a single right-hand side, $\hat{x}$, satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and ϵ is the **machine precision**. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$, the condition number of A with respect to the solution of the linear equations. F04BAF uses the approximation $\|E\|_1 = \epsilon \|A\|_1$ to estimate ERRBND. See Section 4.4 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating-point operations required to solve the equations $AX = B$ is proportional to $(\frac{2}{3}n^3 + n^2r)$. The condition number estimation typically requires between four and five solves and never more than eleven solves, following the factorization.

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

The complex analogue of F04BAF is F04CAF.

9 Example

This example solves the equations

$$AX = B,$$

where

$$A = \begin{pmatrix} 1.80 & 2.88 & 2.05 & -0.89 \\ 5.25 & -2.95 & -0.95 & -3.80 \\ 1.58 & -2.69 & -2.90 & -1.04 \\ -1.11 & -0.66 & -0.59 & 0.80 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 9.52 & 18.47 \\ 24.35 & 2.25 \\ 0.77 & -13.28 \\ -6.22 & -6.21 \end{pmatrix}.$$

An estimate of the condition number of A and an approximate error bound for the computed solutions are also printed.

9.1 Program Text

```

*      F04BAF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, NRHSMX
      PARAMETER        (NMAX=8,NRHSMX=2)
      INTEGER          LDA, LDB
      PARAMETER        (LDA=NMAX,LDB=NMAX)
*      .. Local Scalars ..
      DOUBLE PRECISION ERRBND, RCOND
      INTEGER          I, IERR, IFAIL, J, N, NRHS
*      .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), B(LDB,NRHSMX)
      INTEGER          IPIV(NMAX)
*      .. External Subroutines ..
      EXTERNAL         F04BAF, X04CAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F04BAF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NRHS
      IF (N.LE.NMAX .AND. NRHS.LE.NRHSMX) THEN
*
*          Read A and B from data file
*
*          READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
*          READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*          Solve the equations AX = B for X
*
*          IFAIL = 1
*          CALL F04BAF(N,NRHS,A,LDA,IPIV,B,LDB,RCOND,ERRBND,IFAIL)
*
*          IF (IFAIL.EQ.0) THEN
*
*              Print solution, estimate of condition number and approximate
*              error bound
*
*              IERR = 0
*              CALL X04CAF('General',' ',N,NRHS,B,LDB,'Solution',IERR)
*
*              WRITE (NOUT,*)
*              WRITE (NOUT,*) 'Estimate of condition number'
*              WRITE (NOUT,99999) 1.0D0/RCOND
*              WRITE (NOUT,*)
*              WRITE (NOUT,*)
*              +      'Estimate of error bound for computed solutions'
*              WRITE (NOUT,99999) ERRBND
*              ELSE IF (IFAIL.EQ.N+1) THEN
*
*                  Matrix A is numerically singular. Print estimate of
*                  reciprocal of condition number and solution
*
*                  WRITE (NOUT,*)
*                  WRITE (NOUT,*) 'Estimate of reciprocal of condition number'
*                  WRITE (NOUT,99999) RCOND
*
*                  WRITE (NOUT,*)
*                  IERR = 0
*                  CALL X04CAF('General',' ',N,NRHS,B,LDB,'Solution',IERR)
*
*              ELSE IF (IFAIL.GT.0 .AND. IFAIL.LE.N) THEN
*
*                  The upper triangular matrix U is exactly singular. Print
*                  details of factorization
*
*                  WRITE (NOUT,*)

```

```

      IERR = 0
      CALL X04CAF('General',' ',N,N,A,LDA,
+              'Details of factorization',IERR)
*
*      Print pivot indices
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Pivot indices'
      WRITE (NOUT,99998) (IPIV(I),I=1,N)
      ELSE
      WRITE (NOUT,99997) IFAIL
      END IF
    ELSE
      WRITE (NOUT,*) 'NMAX and/or NRHSMX too small'
    END IF
*
99999 FORMAT (6X,1P,E9.1)
99998 FORMAT ((1X,7I11))
99997 FORMAT (1X,' ** F04BAF returned with IFAIL = ',I5)
      END

```

9.2 Program Data

F04BAF Example Program Data

```

      4          2          :Values of N and NRHS

      1.80      2.88      2.05  -0.89
      5.25     -2.95     -0.95  -3.80
      1.58     -2.69     -2.90  -1.04
      -1.11    -0.66     -0.59   0.80 :End of matrix A

      9.52     18.47
      24.35     2.25
      0.77    -13.28
      -6.22    -6.21          :End of matrix B

```

9.3 Program Results

F04BAF Example Program Results

Solution

	1	2
1	1.0000	3.0000
2	-1.0000	2.0000
3	3.0000	4.0000
4	-5.0000	1.0000

Estimate of condition number

1.5E+02

Estimate of error bound for computed solutions

1.7E-14